POLYTECHNIQUE
MONTRÉAL

Transport solvers available in DRAGON5

A. Hébert

2015/10/21

Outline

Transport solution doors in DRAGON5
The interface current method with SYBILT
The collision probability method with NXT
The method of characteristics with NXT
The collision probability method with SALT
The discrete ordinates method with SNT
Datasets

Transport solution doors in DRAGON5

A transport solution door is a generic API permitting to solve

- a single monoenergetic or
- a set of G independent monoenergetic Boltzmann transport equations (BTE).

The following doors are available:

`SYBILT()`, `SYBILA()`, `SYBILP()`, `SYBILF()` Collision probability method in 1D and interface current method in 2D

`EXCELT()`, `EXCELP()` Collision probability method in Cartesian (2D/3D) or hexagonal 2D.

`NXT()` Collision probability method (new generation) in Cartesian (2D/3D) or hexagonal (2D/3D).

`SALT()` Collision probability method (new generation) in general 2D surfacic geometry.

`MCCGT()`, `MCCGA()`, `MCCGF()` Method of characteristics (MOC) software adaptor for EXCELL, NXT or SALT doors.

`SNT()`, `SNF()` Method of discrete ordinates (SN) in 1D, R-Z (2D) or regular Cartesian geometry (2D,3D).

The interface current method with SYBILT

- The interface current (IC) method is an approximate solution technique of the BTE. Implemented in SYBIL doors.
- The fuel assembly (Cartesian or hexagonal) is divided in pincells
  - 2D collision probabilities are computed over each pincell
  - Pincells are coupled using DP_0 and DP_1 currents between them
- IC method is recommended for performing resonance self-shielding calculations over PWR assemblies
- IC method is used to obtain the solution of the BTE
  - as the main solution technique in previous-generation computational schemes ($\simeq$1995)
  - as the solution technique for the first level in today's two-level schemes
- Two solution techniques:

  `PIJ`: Build a full collision probability matrix for the assembly

  `ARM`: Perform a flux-current iteration in each energy groups

The interface current method with SYBILT

The dataflow of a DRAGON5 dataset with SYBIL looks like



The corresponding dataset is

```
GEOM := GEO: :: CARCEL 6
  X- REFL X+ REFL    Y- REFL Y+ REFL
  MESHX 0.0 1.2  MESHY 0.0 1.2
  RADIUS 0.0 0.288712 0.365195 0.397962
         0.4083 0.45 0.5748331
  MIX 1 2 3 4 5 6 7 ;
TRACK := SYBILT: GEOM ::
  MAXR 20 QUA2 20 3 ;
LIBRARY2 := USS: LIBRARY TRACK ::
  EDIT 1 TRAN PASS 2 PIJ ;
ASB := ASM: LIBRARY2 TRACK :: PIJ ;
FLUX := FLU: LIBRARY2 TRACK ASB :: TYPE K ;
```

The collision probability method with NXT

- The collision probabiility (PIJ) method is a solution technique of the BTE. Implemented in SYBIL (1D only), EXCEL, NXT and SALT doors
- NXT provides the latest implementation of the CP method in 2D/3D Cartesian/hexagonal geometries
- Two types of boundary conditions:
  - `TISO` Finite tracks (`VOID`, `ALBE` or `REFL`-white BCs)
  - `TSPC` Cyclic tracks (`TRAN`, `SYME` or `REFL`-specular BCs)
- A tracking file in sequential binary format is produced



Finite tracks



Cyclic tracks

The collision probability method with NXT

The dataflow of a DRAGON5 dataset with NXT looks like



The corresponding dataset is

```
GEOM := GEO: :: CARCEL 6
  X- REFL X+ REFL    Y- REFL Y+ REFL
  MESHX 0.0 1.2  MESHY 0.0 1.2
  RADIUS 0.0 0.288712 0.365195 0.397962
         0.4083 0.45 0.5748331
  MIX 1 2 3 4 5 6 7 ;
TRACK TRKFL := NXT: GEOM ::
  TISO 20 15.0 ;
LIBRARY2 := USS: LIBRARY TRACK TRKFL ::
  EDIT 1 TRAN PASS 2 PIJ ;
ASB := ASM: LIBRARY2 TRACK TRKFL :: PIJ ;
FLUX := FLU: LIBRARY2 TRACK ASB :: TYPE K ;
```

The collision probability method with NXT

It is possible to avoid computing a tracking file and to compute the track lengths on demand during the calculation of the PIJ matrices. This option (XCLL) increases the CPU time required to compute the PIJ matrices.



The corresponding dataset is

```
GEOM := GEO: :: CARCEL 6
  X- REFL X+ REFL    Y- REFL Y+ REFL
  MESHX 0.0 1.2  MESHY 0.0 1.2
  RADIUS 0.0 0.288712 0.365195 0.397962
         0.4083 0.45 0.5748331
  MIX 1 2 3 4 5 6 7 ;
TRACK := NXT: GEOM ::
  XCLL TISO 20 15.0 ;
LIBRARY2 := USS: LIBRARY TRACK ::
  EDIT 1 TRAN PASS 2 PIJ ;
ASB := ASM: LIBRARY2 TRACK :: PIJ ;
FLUX := FLU: LIBRARY2 TRACK ASB :: TYPE K ;
```

The collision probability method with NXT

It is possible to use module PSP: to produce the postscript representation of a NXT geometry.



The corresponding dataset is

```
SEQ_ASCII Fil_NXT.ps :: FILE './Fil_NXT.ps' ;
*
GEOM := GEO: :: CARCEL 6
  X- REFL X+ REFL    Y- REFL Y+ REFL
  MESHX 0.0 1.2  MESHY 0.0 1.2
  RADIUS 0.0 0.288712 0.365195 0.397962
         0.4083 0.45 0.5748331
  MIX 1 2 3 4 5 6 7 ;
TRACK TRKFL := NXT: GEOM ::
  TISO 20 15.0 ;
Fil_NXT.ps := PSP: TRACK ::
  FILL CMYK
  TYPE REGION ;
```

The collision probability method with NXT

The postscript file generated by `PSP:` follows:



The method of characteristics with NXT

- The method of characteristics (MOC) is a solution technique of the BTE. Available with `EXCELT()`, `NXT()` and `SALT()` doors
- The MOC offers an alternative solution technique to the PIJ method, with the following characteristics
  - Full $N \times N$ matrices (if N regions) are never computed
  - Scattering anisotropy of the neutron sources can be taken into account
  - Discontinuous-linear neutron source distributions can be taken into account
  - Scattering reduction of the monoenergetic BTE is not possible and inner iterations must be implemented with
    - acceleration with the GMRes algorithm
    - preconditionning with a simplified transport operator
- Doors `MCCGT()`, `MCCGA()` and `MCCGF()` are used in addition to `NXT()`

The method of characteristics with NXT

The dataflow of a DRAGON5 dataset with NXT looks like



The corresponding dataset is

```
GEOM := GEO: :: CARCEL 6
  X- REFL X+ REFL    Y- REFL Y+ REFL
  MESHX 0.0 1.2  MESHY 0.0 1.2
  RADIUS 0.0 0.288712 0.365195 0.397962
         0.4083 0.45 0.5748331
  MIX 1 2 3 4 5 6 7 ;
TRACK TRKFL := NXT: GEOM ::
  TISO 20 15.0 ;
TRACK := MCCGT: TRACK TRKFL ::
  LCMD 3 AAC 80 TMT EPSI 1E-5 MCU 10000 ;
LIBRARY2 := USS: LIBRARY TRACK TRKFL ::
  EDIT 1 TRAN PASS 2 ARM ;
ASB := ASM: LIBRARY2 TRACK TRKFL :: ARM ;
FLUX := FLU: LIBRARY2 TRACK TRKFL ASB :: TYPE K ;
```

The collision probability method with SALT

- The SALT door provides another implementation of the PIJ and MOC solutions for the BTE.
- SALT implements a general 2D geometry model based on a surfacic representation. Any 2D geometry is seen as a collection of surfacic nodes of three types:
  - straight line segment
  - arc circles
  - full circles
- Introduced in the Ph.D. thesis of N. Lyoussi-Charrat, CEA,1994
- A first module `G2S:` produces the surfacic file and a second module `SALT:` perform the tracking of this surfacic file
- Module `SALT:` is largely based on the internal API of the `NXT:` module
- The surfacic file can also be produced using the SALOME platform with the help of a Python plugin
- Windmill discretizatization of the Cartesian pincells are expected in DRAGON V5.0.2.

The collision probability method with SALT

The dataflow of a DRAGON5 dataset with SALT looks like



The corresponding dataset is

```
GEOM := GEO: :: CARCEL 6
  X- ALBE 1.0 X+ ALBE 1.0    Y- ALBE 1.0 Y+ ALBE 1.0
  MESHX 0.0 1.2  MESHY 0.0 1.2
  RADIUS 0.0 0.288712 0.365195 0.397962
         0.4083 0.45 0.5748331
  MIX 1 2 3 4 5 6 7 ;
Fil_SAL.dat Fil_SAL.ps := G2S: GEOM :: DRAWNOD ;
TRACK TRKFL := SALT: Fil_SAL.dat ::
  TISO 20 15.0 ;
LIBRARY2 := USS: LIBRARY TRACK TRKFL ::
  EDIT 1 TRAN PASS 2 PIJ ;
ASB := ASM: LIBRARY2 TRACK TRKFL :: PIJ ;
FLUX := FLU: LIBRARY2 TRACK ASB :: TYPE K ;
```

The collision probability method with SALT

The surfacic and postscript files are declared as

```
SEQ_ASCII Fil_SAL.dat ;
SEQ_ASCII Fil_SAL.ps :: FILE './Fil_SAL.ps' ;
```

The postscript file generated by `G2S:` follows:



The collision probability method with SALT

The G2S: module permits windmill discretization of Cartesian pincells in a fuel assembly (as of DRAGON 5.0.2)



The collision probability method with SALT

It is possible to use module TLM: to print individual tracks (with the help of Matlab). Note keyword LONG in module SALT:.

```
SEQ_ASCII Fil_SAL.dat ;
SEQ_ASCII Lines.m :: FILE './Lines_tiso.m' ;
*
GEOM := GEO: :: CARCEL 6
  X- ALBE 1.0 X+ ALBE 1.0    Y- ALBE 1.0 Y+ ALBE 1.0
  MESHX 0.0 1.2  MESHY 0.0 1.2
  RADIUS 0.0 0.288712 0.365195 0.397962
         0.4083 0.45 0.5748331
  MIX 1 2 3 4 5 6 7 ;
Fil_SAL.dat := G2S: GEOM ;
TRACK TRKFL := SALT: Fil_SAL.dat ::
  TISO 3 3.0 LONG ;

Lines.m := TLM: TRACK TRKFL ::
  NTPO 1
  DIRECTIONS NoPause DIR 0 ;
```

The collision probability method with SALT

The `./Lines_tiso.m` file is recovered at the end of the DRAGON5 run and is executed in Matlab. The postscript file generated by Matlab follows:



The discrete ordinates method with SNT

- The discrete ordinates (or SN) method is available in DRAGON5. The basic implementation is the diamond differencing (DD) scheme with no negative fixup.
- Solution is possible in
  - 1D slab, cylindrical and spherical geometries
  - RZ regular geometry
  - 1D/2D/3D regular Cartesian geometries
- High order diamond differencing (HODD) schemes are available in Cartesian geometries.
- Scattering anisotropy of the neutron sources can be taken into account
- A choice of angular quadratures is available, including level symmetric and product quadratures.
- Pincell geometries cannot be represented. Cylinderization is required.

The discrete ordinates method with SNT

The dataflow of a DRAGON5 dataset with SNT looks like



The corresponding dataset is

```
GEOM := GEO: :: TUBE 7
  R+ REFL
  RADIUS 0.0 0.288712 0.365195 0.397962
         0.4083 0.45 0.5748331 0.6770275
  MIX 1 2 3 4 5 6 7 ;
TRACK := SNT: GEOM ::
  EDIT 1 SN 12 SCAT 1 QUAD 1 ;
LIBRARY2 := USS: LIBRARY TRACK ::
  EDIT 1 TRAN PASS 2 ARM ;
ASB := ASM: LIBRARY2 TRACK :: ARM ;
FLUX := FLU: LIBRARY2 TRACK ASB :: TYPE K ;
```

Datasets

Complete DRAGON5 datasets are provided for the above workshop examples

Transport solution	K_{eff}	DRAGON5 dataset
SYBILT:	1.331530	workshop_sybil.x2m
NXT:	1.330710	workshop_nxt.x2m
MCCGT:	1.330929	workshop_mccg.x2m
SALT:	1.331824	workshop_salt.x2m
SNT:	1.329972	workshop_sn.x2m